CURVAS

Curvas PLANAS

■ Traza de curvas planas en forma implícita

Comenzamos con un trazado básico de una elipse.
Ejercicio: Calcula su centro y dibuja sus ejes.

```
In[233]:=
        Clear["Global`*"]
        Needs["Graphics`ImplicitPlot`"]
        f[x_, y_] := x^2 + 4 y^2 - 4 x - 8 y + 7
        ImplicitPlot[f[x, y] == 0, {x, 0, 4}, {y, 0, 2}]
        ("{x,0,4},{y,0,2} define el rango máximo de los ejes coordenados.")
```



Completando cuadrados, observamos que f[x, y] = (x - 2)^2 + 4 (y - 1)^2 - 1. Pintamos los ejes.

```
In[6]:=  ((x - 2)^2 + 4 (y - 1)^2 - 1) - f[x, y] // Expand

Out[6]=  0

In[7]:=  ImplicitPlot[{f[x, y] == 1, x == 2, y == 1}, {x, 0, 4}, {y, -0, 2}]
```



Veamos algunas variantes.
Thickness[0.005] da grosor a la curva.
Dashing[{0.04,0.02}] dibuja la curva a trazos, el primer número indica la longitud de los trazos negros.

```
In[8]:= Clear["Global`*"]
        f[x_, y_] := x^2 + y^2
        g[x_, y_] := x - 4 y^2
        ImplicitPlot[{f[x, y] == 1, g[x, y] == 0}, {x, -1.5, 1.5}, {y, -1, 1},
         PlotStyle → {{Thickness[0.005], Dashing[{0.04, 0.02}]}, {Thickness[0.019]}}]
```



■ Traza de curvas planas en paramétricas

Comenzamos con un trazado básico.

```
In[12]:= Clear["Global`*"]
         alpha1[t_] := {4 E^(-t) Cos[3 t], 4 E^(-t) Sin[3 t]}
         ParametricPlot[alpha1[t], {t, 0, Pi} ]
```




```
Out[14]= - Graphics -
```

Observamos su movimiento, fijamos la ventana con PlotRange→{{-2,4},{-1,3}} y dibujamos intervalos de tiempo cada vez mayores.
Prueba a hacerlo con Do o For.

```
In[15]:= ParametricPlot[alpha1[t], {t, 0, 1}, PlotRange → {{-2, 4}, {-1, 3}}];
         ParametricPlot[alpha1[t], {t, 0, 2}, PlotRange → {{-2, 4}, {-1, 3}}];
         ParametricPlot[alpha1[t], {t, 0, 3}, PlotRange → {{-2, 4}, {-1, 3}}];
         ParametricPlot[alpha1[t], {t, 0, 4}, PlotRange → {{-2, 4}, {-1, 3}}];
```









Veamos algunas variantes.
Thickness[0.005] da grosor a la curva.
Dashing[{0.04,0.02}] dibuja la curva a trazos, el primer número indica la longitud de los trazos negros.

```
In[19]:= Clear["Global`*"]

         alpha1[t_] := {E^(-t) * Cos[3 t], E^(-t) * Sin[3 t]}
         alpha2[t_] := {-E^(-t) * Sin[3 t], E^(-t) * Cos[3 t]}
         ParametricPlot[Evaluate[{alpha1[t], alpha2[t]}],
           {t, 1, 4 Pi}, AspectRatio → Automatic ];
         ParametricPlot[{alpha1[t], alpha2[t]}, {t, 1, 4 Pi} , PlotRange → All];
         ParametricPlot[{alpha1[t], alpha2[t]}, {t, 1, 4 Pi} ,
           PlotRange → {{-.1, 0.1}, {-.05, 0.15}}, AspectRatio → Automatic];
```


0.01
0.005
-0.01
-0.005
0.005
0.01
-0.005
-0.01


0.1
-0.3
-0.2
-0.1
0.1
0.2
-0.1
-0.2
-0.3


0.125
0.1
0.075
0.05
0.025
-0.1
-0.075
-0.05
-0.025
0.025
0.05
0.075
0.1
-0.025
-0.05

■ Vector tangente y normal

```
In[25]:= Clear["Global`*"]
         alpha[t_] := (Sin[3 t] + 4) * {Cos[t], Sin[t]}
         traza = ParametricPlot[alpha[t], {t, -Pi, Pi}, PlotStyle -> Dashing[{0.02}]]
```




```
Out[27]= - Graphics -
```

Vector tangente y vector normal (previa comprobación de que N(t) apunta en la misma dirección que T'(t)).

```
In[28]:= T1[t_] := alpha'[t] / Sqrt[alpha'[t].alpha'[t]] // Simplify
         N1[t_] := T1'[t] / Sqrt[T1'[t].T1'[t]] // Simplify
         T1[t].N1[t] // Simplify

Out[30]= 0
```

Representaciones gráficas

```
In[31]:= << Graphics`PlotField`
         tangentes = ListPlotVectorField[
            Table[{alpha[i], T1[i]}, {i, -Pi, Pi, 0.4}], AspectRatio → Automatic];
         normales = ListPlotVectorField[Table[{alpha[i], N1[i]}, {i, -Pi, Pi, 0.4}],
            AspectRatio → Automatic];
         diedro = Show[traza, tangentes, normales]
```




```
Out[34]= - Graphics -
```

■ Curvatura PLANA (usando Module)

```
In[35]:= Clear["Global`*"]
         alpha[t_] := (Sin[3 t] + 4) * {Cos[t], Sin[t]};
```

Para la curvatura, nótese:
sabemos que dT/ds = k N,   y que ds/dt = $|\alpha'|$
entonces
dT/dt = k N $|\alpha'|$
y tomando módulos
k = |dT/dt| / $|\alpha'|$

```
In[37]:= curvatura2D[alpha_, t_] := Module[{T1, k},
           T1[t] = D[alpha[t], t] / Sqrt[D[alpha[t], t].D[alpha[t], t]];
           k =
            Sqrt[D[T1[t], t].D[T1[t], t]] / Sqrt[D[alpha[t], t].D[alpha[t], t]] // Simplify;
           Return[k]]
         curvatura2D[alpha, t]
```

Out[38]= $$\frac{2\sqrt{\frac{(15+2\,\mathrm{Cos}[6\,t]+22\,\mathrm{Sin}[3\,t])^2}{(21+4\,\mathrm{Cos}[6\,t]+8\,\mathrm{Sin}[3\,t])^2}}}{\sqrt{21+4\,\mathrm{Cos}[6\,t]+8\,\mathrm{Sin}[3\,t]}}$$

```
In[39]:= Plot[Evaluate[curvatura2D[alpha, t]], {t, -Pi, Pi}]
```



■ **Determinar una curva plana a partir su ecuación intrínseca**

Se trata de encontrar una curva p.p.a. cuya curvatura en función de la longitud de arco s sea k(s).
Como k(s)=$|\alpha''(s)|$, se trata, más o menos, de integrar dos veces.
Primero se define $\theta(s)=\int k(s)\,ds$ y se busca α tal que $\alpha'(s)=(\cos(\theta(s)),\mathrm{sen}(\theta(s)))$ que, obviamente, tiene norma 1 y cumple $|\alpha''(s)=k(s)$. Por lo tanto la curva plana buscada es $\alpha(s)=\int(\cos(\theta(s)),\ \mathrm{sen}(\theta(s)))\,ds$ es la curva buscada.
En el ejemplo, además, lo haremos en función de un parámetro: a.

```
In[174]:=
        Clear["Global`*"]
        k = 1 / Sqrt[2 a s];
        theta = Integrate[k, s] // PowerExpand
        alpha := Integrate[{Cos[theta], Sin[theta]}, s] // Simplify
        Print["La curva con curvatura", k, " es α(s) = ", alpha]
```

Out[176]=
$$\frac{\sqrt{2}\,\sqrt{s}}{\sqrt{a}}$$

La curva con curvatura $\frac{1}{\sqrt{2}\,\sqrt{a\,s}}$ es α(s) =

$$\left\{a\,\mathrm{Cos}\left[\frac{\sqrt{2}\,\sqrt{s}}{\sqrt{a}}\right]+\sqrt{2}\,\sqrt{a}\,\sqrt{s}\,\mathrm{Sin}\left[\frac{\sqrt{2}\,\sqrt{s}}{\sqrt{a}}\right],\ -\sqrt{2}\,\sqrt{a}\,\sqrt{s}\,\mathrm{Cos}\left[\frac{\sqrt{2}\,\sqrt{s}}{\sqrt{a}}\right]+a\,\mathrm{Sin}\left[\frac{\sqrt{2}\,\sqrt{s}}{\sqrt{a}}\right]\right\}$$

Verificación

In[185]:=
```
D[alpha, s].D[alpha, s] // Simplify
Sqrt[D[alpha, {s, 2}].D[alpha, {s, 2}]] // Simplify
```

Out[185]=
1

Out[186]=
$$\frac{\sqrt{\frac{1}{a\,s}}}{\sqrt{2}}$$

Representación para a=3.

In[191]:=
```
a = 3;
ParametricPlot[Evaluate[alpha], {s, 0.001, 35}, AspectRatio → Automatic]
```



Curvas en el ESPACIO

■ Curvas parametrizadas

In[15]:=
```
Clear["Global`*"]
alpha[t_] := {r * Cos[t / r], r * Sin[t / r], 2 r Sin[t / (2 r)]}
```

■ Comprobación de que la curva está contenida en sendas superficies

Veamos que la curva está contenide en el cilindro $x^2 + y^2 = r^2$ y en la esfera $x^2 + y^2 + z^2 = 4 r^2$



```
In[3]:= Simplify[x^2 + y^2 - r^2 /. {x → alpha[t][[1]], y → alpha[t][[2]], z → alpha[t][[3]]}]
        Simplify[
         (x + r)^2 + y^2 + z^2 - 4 r^2 /. {x → alpha[t][[1]], y → alpha[t][[2]], z → alpha[t][[3]]}]
```

Out[3]= 0

Out[4]= 0

■ Derivación

```
In[54]:= D[alpha[t], t]
         alpha'[t]
         D[alpha[t], {t, 2}]
         alpha''[t]
```

Out[54]= $\left\{-\mathrm{Sin}\left[\frac{t}{r}\right], \mathrm{Cos}\left[\frac{t}{r}\right], \mathrm{Cos}\left[\frac{t}{2\,r}\right]\right\}$

Out[55]= $\left\{-\mathrm{Sin}\left[\frac{t}{r}\right], \mathrm{Cos}\left[\frac{t}{r}\right], \mathrm{Cos}\left[\frac{t}{2\,r}\right]\right\}$

Out[56]= $\left\{-\frac{\mathrm{Cos}\left[\frac{t}{r}\right]}{r}, -\frac{\mathrm{Sin}\left[\frac{t}{r}\right]}{r}, -\frac{\mathrm{Sin}\left[\frac{t}{2\,r}\right]}{2\,r}\right\}$

Out[57]= $\left\{-\frac{\mathrm{Cos}\left[\frac{t}{r}\right]}{r}, -\frac{\mathrm{Sin}\left[\frac{t}{r}\right]}{r}, -\frac{\mathrm{Sin}\left[\frac{t}{2\,r}\right]}{2\,r}\right\}$

■ Producto escalar (.) , norma (Norm), producto vectorial (Cross o ×) y mixto (Det)

In[58]:=
```
Simplify[alpha'[t].alpha'[t]]
Norm[alpha'[t]] // Expand

Simplify[Cross[alpha'[t], alpha''[t]]]
Simplify[alpha'[t] × alpha''[t]]

Simplify[(alpha'[t] × alpha''[t]).alpha'''[t]]
Simplify[Det[{alpha'[t], alpha''[t], alpha'''[t]}]]
```

Out[58]= $\frac{1}{2}\left(3+\mathrm{Cos}\left[\frac{t}{r}\right]\right)$

Out[59]= $\sqrt{\mathrm{Abs}\left[\mathrm{Cos}\left[\frac{t}{2r}\right]\right]^2+\mathrm{Abs}\left[\mathrm{Cos}\left[\frac{t}{r}\right]\right]^2+\mathrm{Abs}\left[\mathrm{Sin}\left[\frac{t}{r}\right]\right]^2}$

Out[60]= $\left\{\frac{3\,\mathrm{Sin}\left[\frac{t}{2r}\right]+\mathrm{Sin}\left[\frac{3t}{2r}\right]}{4\,r},-\frac{\mathrm{Cos}\left[\frac{t}{2r}\right]^3}{r},\frac{1}{r}\right\}$

Out[61]= $\left\{\frac{3\,\mathrm{Sin}\left[\frac{t}{2r}\right]+\mathrm{Sin}\left[\frac{3t}{2r}\right]}{4\,r},-\frac{\mathrm{Cos}\left[\frac{t}{2r}\right]^3}{r},\frac{1}{r}\right\}$

Out[62]= $\frac{3\,\mathrm{Cos}\left[\frac{t}{2r}\right]}{4\,r^3}$

Out[63]= $\frac{3\,\mathrm{Cos}\left[\frac{t}{2r}\right]}{4\,r^3}$

■ Triedro de Frenet

In[64]:=
```
T1 = Simplify[alpha'[t] / Sqrt[alpha'[t].alpha'[t]]]
aux = Cross[alpha'[t], alpha''[t]] // Simplify;
B1 = PowerExpand[aux / Sqrt[aux.aux]] // Simplify
N1 = Cross[B1, T1]
```

Out[64]= $\left\{-\frac{\sqrt{2}\,\mathrm{Sin}\left[\frac{t}{r}\right]}{\sqrt{3+\mathrm{Cos}\left[\frac{t}{r}\right]}},\frac{\sqrt{2}\,\mathrm{Cos}\left[\frac{t}{r}\right]}{\sqrt{3+\mathrm{Cos}\left[\frac{t}{r}\right]}},\frac{\sqrt{2}\,\mathrm{Cos}\left[\frac{t}{2r}\right]}{\sqrt{3+\mathrm{Cos}\left[\frac{t}{r}\right]}}\right\}$

Out[66]= $\left\{\frac{3\,\mathrm{Sin}\left[\frac{t}{2r}\right]+\mathrm{Sin}\left[\frac{3t}{2r}\right]}{r\sqrt{\frac{26+6\,\mathrm{Cos}\left[\frac{t}{r}\right]}{r^2}}},-\frac{4\,\mathrm{Cos}\left[\frac{t}{2r}\right]^3}{r\sqrt{\frac{26+6\,\mathrm{Cos}\left[\frac{t}{r}\right]}{r^2}}},\frac{4}{r\sqrt{\frac{26+6\,\mathrm{Cos}\left[\frac{t}{r}\right]}{r^2}}}\right\}$

Out[67]= $\Bigg\{-\frac{4\sqrt{2}\,r\,\mathrm{Cos}\left[\frac{t}{2r}\right]^4\sqrt{\frac{26+6\,\mathrm{Cos}\left[\frac{t}{r}\right]}{r^2}}}{\sqrt{3+\mathrm{Cos}\left[\frac{t}{r}\right]}\left(26+6\,\mathrm{Cos}\left[\frac{t}{r}\right]\right)}-\frac{4\sqrt{2}\,r\,\mathrm{Cos}\left[\frac{t}{r}\right]\sqrt{\frac{26+6\,\mathrm{Cos}\left[\frac{t}{r}\right]}{r^2}}}{\sqrt{3+\mathrm{Cos}\left[\frac{t}{r}\right]}\left(26+6\,\mathrm{Cos}\left[\frac{t}{r}\right]\right)},$

$-\frac{3\sqrt{2}\,r\,\mathrm{Cos}\left[\frac{t}{2r}\right]\sqrt{\frac{26+6\,\mathrm{Cos}\left[\frac{t}{r}\right]}{r^2}}\,\mathrm{Sin}\left[\frac{t}{2r}\right]}{\sqrt{3+\mathrm{Cos}\left[\frac{t}{r}\right]}\left(26+6\,\mathrm{Cos}\left[\frac{t}{r}\right]\right)}-\frac{4\sqrt{2}\,r\sqrt{\frac{26+6\,\mathrm{Cos}\left[\frac{t}{r}\right]}{r^2}}\,\mathrm{Sin}\left[\frac{t}{r}\right]}{\sqrt{3+\mathrm{Cos}\left[\frac{t}{r}\right]}\left(26+6\,\mathrm{Cos}\left[\frac{t}{r}\right]\right)}-$

$\frac{\sqrt{2}\,r\,\mathrm{Cos}\left[\frac{t}{2r}\right]\sqrt{\frac{26+6\,\mathrm{Cos}\left[\frac{t}{r}\right]}{r^2}}\,\mathrm{Sin}\left[\frac{3t}{2r}\right]}{\sqrt{3+\mathrm{Cos}\left[\frac{t}{r}\right]}\left(26+6\,\mathrm{Cos}\left[\frac{t}{r}\right]\right)},\frac{3\sqrt{2}\,r\,\mathrm{Cos}\left[\frac{t}{r}\right]\sqrt{\frac{26+6\,\mathrm{Cos}\left[\frac{t}{r}\right]}{r^2}}\,\mathrm{Sin}\left[\frac{t}{2r}\right]}{\sqrt{3+\mathrm{Cos}\left[\frac{t}{r}\right]}\left(26+6\,\mathrm{Cos}\left[\frac{t}{r}\right]\right)}-$

$\frac{4\sqrt{2}\,r\,\mathrm{Cos}\left[\frac{t}{2r}\right]^3\sqrt{\frac{26+6\,\mathrm{Cos}\left[\frac{t}{r}\right]}{r^2}}\,\mathrm{Sin}\left[\frac{t}{r}\right]}{\sqrt{3+\mathrm{Cos}\left[\frac{t}{r}\right]}\left(26+6\,\mathrm{Cos}\left[\frac{t}{r}\right]\right)}+\frac{\sqrt{2}\,r\,\mathrm{Cos}\left[\frac{t}{r}\right]\sqrt{\frac{26+6\,\mathrm{Cos}\left[\frac{t}{r}\right]}{r^2}}\,\mathrm{Sin}\left[\frac{3t}{2r}\right]}{\sqrt{3+\mathrm{Cos}\left[\frac{t}{r}\right]}\left(26+6\,\mathrm{Cos}\left[\frac{t}{r}\right]\right)}\Bigg\}$

```
In[68]:= T1 /. {t → Pi / 2, r → 1}
         N1 /. {t → Pi / 2, r → 1}
         B1 /. {t → Pi / 2, r → 1}
```

Out[68]= $\left\{-\sqrt{\frac{2}{3}}, 0, \frac{1}{\sqrt{3}}\right\}$

Out[69]= $\left\{-\frac{1}{\sqrt{39}}, -2\sqrt{\frac{3}{13}}, -\sqrt{\frac{2}{39}}\right\}$

Out[70]= $\left\{\frac{2}{\sqrt{13}}, -\frac{1}{\sqrt{13}}, 2\sqrt{\frac{2}{13}}\right\}$

Y como función (escribimos sólo T)

```
In[71]:= Clear["T1"]
         T1[t_] := Simplify[alpha'[t] / Sqrt[alpha'[t].alpha'[t]]]
```

■ Buscando puntos con torsión nula (α'[t]×α''[t]).α'''[t]=0)

```
In[73]:= ss = Solve[Det[{alpha'[t], alpha''[t], alpha'''[t]}] == 0, t]
```

Out[73]= {{t → -π r}, {t → π r}}

```
In[74]:= Simplify[alpha[t]] /. ss
```

Out[74]= {{-r, 0, -2 r}, {-r, 0, 2 r}}

■ Traza de una curva parametrizada en el espacio R3

```
In[5]:= Clear[r]
        r = 1;
        traza = ParametricPlot3D[Evaluate[alpha[t]], {t, 0, 4 Pi r}, ViewPoint → {2.5, 2, 1}]
```



■ Longitud de un arco de curva

```
In[17]:= Clear[r];
         s[t_] := Integrate[Sqrt[alpha'[t].alpha'[t]], t]
         s[t]
```

Out[19]= $2\sqrt{2}\, r\, \text{EllipticE}\left[\frac{t}{2 r}, \frac{1}{2}\right]$

```
In[20]:= r = 1;
         NIntegrate[Sqrt[alpha'[t].alpha'[t]], {t, 0, 4 Pi r}]

Out[21]= 15.2808
```

¿Parametrizamos por el arco? (No siempre)

```
In[22]:= r = 1;
         Solve[s[t] == s, t]

Solve::ifun : Inverse functions are being used by Solve, so some
  solutions may not be found; use Reduce for complete solution information. More…
```

Out[23]= $\left\{\left\{t \to 2\ \text{InverseFunction}[\text{EllipticE}, 1, 2]\left[\frac{s}{2\sqrt{2}}, \frac{1}{2}\right]\right\}\right\}$

■ Parametrizando por el arco

Veamos un ejemplo en que sí es posible parametrizar por el arco, una hélice.

```
In[85]:= alpha[t_] := {Cos[t], Sin[t], t / 3};
         rtrayectoria = ParametricPlot3D[alpha[t], {t, -4 Pi, 4 Pi}]
```




```
In[87]:= s[t_] := Integrate[Sqrt[alpha'[z].alpha'[z]], {z, -4 Pi, t}]
         s[t]
         sol = Solve[s[t] == s, t]
```

Out[88]= $\frac{1}{3}\sqrt{10}\ (4\pi + t)$

Out[89]= $\left\{\left\{t \to -\frac{4\sqrt{10}\ \pi - 3\ s}{\sqrt{10}}\right\}\right\}$

Construimos la reparametrización p.p.a. de la curva:

```
In[90]:= alphappa = alpha[t /. sol[[1]]]
         D[alphappa, s].D[alphappa, s] // Simplify
```

Out[90]= $\left\{\text{Cos}\left[\frac{4\sqrt{10}\ \pi - 3\ s}{\sqrt{10}}\right], -\text{Sin}\left[\frac{4\sqrt{10}\ \pi - 3\ s}{\sqrt{10}}\right], -\frac{4\sqrt{10}\ \pi - 3\ s}{3\sqrt{10}}\right\}$

Out[91]= 1

■ Haz de curvas

■ Dibujo de familias de curvas

Dibujo de curvas de la familia de parámetro r: α_r (t) =(a+r cos(t),r sen(t)) con a =r/sen(p) y p=π/6.
El primer valor de r es -5, el último es 5, incrementándose en cada paso 0.5.
¿Quiénes son las rectas añadidas en el segundo dibujo?

```
In[92]:= Clear["Global`*"]
         p = Pi/6; a = r/Sin[p];
         Show[
          Table[ParametricPlot[{a + r Cos[t], r Sin[t]}, {t, 0, 2 Pi},
            DisplayFunction → Identity], {r, -5, 5, .5}], DisplayFunction → $DisplayFunction]
```




```
In[95]:= Show[
          Table[ParametricPlot[{a + r Cos[t], r Sin[t]},
            {t, 0, 2 Pi}, DisplayFunction → Identity], {r, -5, 5, .5}],
          ParametricPlot[{t, t Tan[p]}, {t, -15, 15}, PlotStyle → {Thickness[0.005]},
           DisplayFunction → Identity], ParametricPlot[{t, -t Tan[p]}, {t, -15, 15},
           PlotStyle → {Thickness[0.005]}, DisplayFunction → Identity],
          DisplayFunction → $DisplayFunction, AspectRatio → Automatic]
```



■ Otros cálculos con curvas parametrizadas

■ Plano osculador

Usamos que los puntos P del plano osculador en α (t) cumplen (P - α (t)) es combinación de α' (t) y α'' (t), para obtener su ecuación implícita.

```
In[96]:= Clear["Global`*"]
         alpha[t_] := {3 t, 3 t^2, 2 t^3}
         X = {x, y, z};
         planoosculador = Det[{(X - alpha[t]), alpha'[t], alpha''[t]}] == 0
```

Out[99]= $-36 t^3 + 36 t^2 x - 36 t y + 18 z == 0$

■ Tecta tangente

In[100]:=
```
Clear["Global`*"]
alpha[t_] := {3 t, 3 t^2, 2 t^3};
rt[t_, u_] := α[t] + u * α'[t]
```

■ Curvatura, torsión, radio de curvatura

In[103]:=
```
Clear["Global`*"]
alpha[t_] := {r * Cos[t], r * Sin[t], 4 t}
norma[z_] := Sqrt[z.z];
```

In[106]:=
```
curvatura[t_] :=
 norma[Cross[alpha'[t], alpha''[t]]] / norma[alpha'[t]]^3 // Simplify
torsion[t_] := -Det[{alpha'[t], alpha''[t], alpha'''[t]}] /
   norma[Cross[alpha'[t], alpha''[t]]]^2 // Simplify
("recordar la nota sobre el signo de la torsión")
radio[t_] := 1 / curvatura[t]
curvatura[t] // PowerExpand
torsion[t]
```

Out[108]=
```
recordar la nota sobre el signo de la torsión
```

Out[110]=

$$\frac{r}{16 + r^2}$$

Out[111]=

$$-\frac{4}{16 + r^2}$$

■ Centro de curvatura y Evoluta

In[112]:=
```
Clear["Global`*"]
alpha[t_] := {Cos[t], Sin[t], t / 2};
norma[z_] := Sqrt[z.z];

T1[t_] := Simplify[alpha'[t] / norma[alpha'[t]]]
B1[t_] :=
  Cross[alpha'[t], alpha''[t]] / norma[Cross[alpha'[t], alpha''[t]]] // Simplify;
N1[t_] := Cross[B1[t], T1[t]]

curvatura[t_] :=
 norma[Cross[alpha'[t], alpha''[t]]] / norma[alpha'[t]]^3 // Simplify
radio[t_] := 1 / curvatura[t]

evoluta[t_] := alpha[t] + N1[t] / curvatura[t]
evoluta[t]
```

Out[121]=

$$\left\{-\frac{\mathrm{Cos}[t]}{4}, -\frac{\mathrm{Sin}[t]}{4}, \frac{t}{2}\right\}$$

In[122]:=
```
ParametricPlot3D[Evaluate[{alpha[t], evoluta[t]}],
 {t, 0, 4 Pi}, ViewPoint → {2.5, 2, 1}]
```



■ Centro de curvatura y Evoluta 2D (usando la 3D)

En este caso, dibujamos el astroide, como evoluta de una elipse.

In[123]:=
```
Clear["Global`*"]
alpha[t_] := {2 * Cos[t], Sin[t], 0};
norma[z_] := Sqrt[z.z];

T1[t_] := Simplify[alpha'[t] / norma[alpha'[t]]]
B1[t_] :=
  Cross[alpha'[t], alpha''[t]] / norma[Cross[alpha'[t], alpha''[t]]] // Simplify;
N1[t_] := Cross[B1[t], T1[t]] // Simplify

curvatura[t_] :=
 norma[Cross[alpha'[t], alpha''[t]]] / norma[alpha'[t]]^3 // Simplify
radio[t_] := 1 / curvatura[t]

evoluta[t_] := alpha[t] + N1[t] / curvatura[t]
evoluta[t]
```

Out[132]=

$$\left\{2\,\mathrm{Cos}[t] - \frac{1}{2}\,\mathrm{Cos}[t]\,(\mathrm{Cos}[t]^2 + 4\,\mathrm{Sin}[t]^2),\ \mathrm{Sin}[t] - \mathrm{Sin}[t]\,(\mathrm{Cos}[t]^2 + 4\,\mathrm{Sin}[t]^2),\ 0\right\}$$

In[133]:=

```
ParametricPlot3D[Evaluate[{alpha[t], evoluta[t]}],
 {t, 0, 4 Pi}, ViewPoint → {0, 0, 1}]
```